// File: ERC20.sol




pragma solidity ^0.5.7;

import "./IERC20.sol";
import "./SafeMath.sol";

/**
 * @title Standard ERC20 token
 *
 * @dev Implementation of the basic standard token.
 * https://eips.ethereum.org/EIPS/eip-20
 * Originally based on code by FirstBlood:
 * https://github.com/Firstbloodio/token/blob/master/smart_contract/FirstBloodToken.sol
 *
 * This implementation emits additional Approval events, allowing applications to reconstruct the allowance status for
 * all accounts just by listening to said events. Note that this isn't required by the specification, and other
 * compliant implementations may not do it.
 */
contract ERC20 is IERC20 {
    using SafeMath for uint256;

    mapping (address => uint256) private _balances;

    mapping (address => mapping (address => uint256)) private _allowed;

    uint256 private _totalSupply;

    /**
     * @dev Total number of tokens in existence.
     */
    function totalSupply() public view returns (uint256) {
        return _totalSupply;
    }

    /**
     * @dev Gets the balance of the specified address.
     * @param owner The address to query the balance of.
     * @return A uint256 representing the amount owned by the passed address.
     */
    function balanceOf(address owner) public view returns (uint256) {
        return _balances[owner];
    }

    /**
     * @dev Function to check the amount of tokens that an owner allowed to a spender.
     * @param owner address The address which owns the funds.
     * @param spender address The address which will spend the funds.
     * @return A uint256 specifying the amount of tokens still available for the spender.
     */
    function allowance(address owner, address spender) public view returns (uint256) {
        return _allowed[owner][spender];
    }

    /**
     * @dev Transfer token to a specified address.
     * @param to The address to transfer to.
     * @param value The amount to be transferred.
     */
    function transfer(address to, uint256 value) public returns (bool) {
        _transfer(msg.sender, to, value);
        return true;
    }

    /**
     * @dev Approve the passed address to spend the specified amount of tokens on behalf of msg.sender.
     * Beware that changing an allowance with this method brings the risk that someone may use both the old
     * and the new allowance by unfortunate transaction ordering. One possible solution to mitigate this
     * race condition is to first reduce the spender's allowance to 0 and set the desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     * @param spender The address which will spend the funds.
     * @param value The amount of tokens to be spent.
     */
    function approve(address spender, uint256 value) public returns (bool) {
        require(value== 0 || _allowed[msg.sender][spender] == 0);
        _approve(msg.sender, spender, value);
        return true;
    }

    /**
     * @dev Transfer tokens from one address to another.
     * Note that while this function emits an Approval event, this is not required as per the specification,
     * and other compliant implementations may not emit the event.
     * @param from address The address which you want to send tokens from
     * @param to address The address which you want to transfer to
     * @param value uint256 the amount of tokens to be transferred
     */
    function transferFrom(address from, address to, uint256 value) public returns (bool) {
        _transfer(from, to, value);
        _approve(from, msg.sender, _allowed[from][msg.sender].sub(value));
        return true;
    }

    /**
     * @dev Increase the amount of tokens that an owner allowed to a spender.
     * approve should be called when _allowed[msg.sender][spender] == 0. To increment
     * allowed value is better to use this function to avoid 2 calls (and wait until
     * the first transaction is mined)
     * From MonolithDAO Token.sol
     * Emits an Approval event.
     * @param spender The address which will spend the funds.
     * @param addedValue The amount of tokens to increase the allowance by.
     */
    function increaseAllowance(address spender, uint256 addedValue) public returns (bool) {
        _approve(msg.sender, spender, _allowed[msg.sender][spender].add(addedValue));
        return true;
    }

    /**
     * @dev Decrease the amount of tokens that an owner allowed to a spender.
     * approve should be called when _allowed[msg.sender][spender] == 0. To decrement
     * allowed value is better to use this function to avoid 2 calls (and wait until
     * the first transaction is mined)
     * From MonolithDAO Token.sol
     * Emits an Approval event.
     * @param spender The address which will spend the funds.
     * @param subtractedValue The amount of tokens to decrease the allowance by.
     */
    function decreaseAllowance(address spender, uint256 subtractedValue) public returns (bool) {
        _approve(msg.sender, spender, _allowed[msg.sender][spender].sub(subtractedValue));
        return true;
    }

    /**
     * @dev Transfer token for a specified addresses.
     * @param from The address to transfer from.
     * @param to The address to transfer to.
     * @param value The amount to be transferred.
     */
    function _transfer(address from, address to, uint256 value) internal {
        require(to != address(0), "ERC20: transfer to the zero address");

        _balances[from] = _balances[from].sub(value);
        _balances[to] = _balances[to].add(value);
        emit Transfer(from, to, value);
    }

    /**
     * @dev Internal function that mints an amount of the token and assigns it to
     * an account. This encapsulates the modification of balances such that the
     * proper events are emitted.
     * @param account The account that will receive the created tokens.
     * @param value The amount that will be created.
     */
    function _mint(address account, uint256 value) internal {
        require(account != address(0), "ERC20: mint to the zero address");

        _totalSupply = _totalSupply.add(value);
        _balances[account] = _balances[account].add(value);
        emit Transfer(address(0), account, value);
    }

    /**
     * @dev Internal function that burns an amount of the token of a given
     * account.
     * @param account The account whose tokens will be burnt.
     * @param value The amount that will be burnt.
     */
    function _burn(address account, uint256 value) internal {
        require(account != address(0), "ERC20: burn from the zero address");

        _totalSupply = _totalSupply.sub(value);
        _balances[account] = _balances[account].sub(value);
        emit Transfer(account, address(0), value);
    }

    /**
     * @dev Approve an address to spend another addresses' tokens.
     * @param owner The address that owns the tokens.
     * @param spender The address that will spend the tokens.
     * @param value The number of tokens that can be spent.
     */
    function _approve(address owner, address spender, uint256 value) internal {
        require(owner != address(0), "ERC20: approve from the zero address");
        require(spender != address(0), "ERC20: approve to the zero address");

        _allowed[owner][spender] = value;
        emit Approval(owner, spender, value);
    }

    /**
     * @dev Internal function that burns an amount of the token of a given
     * account, deducting from the sender's allowance for said account. Uses the
     * internal burn function.
     * Emits an Approval event (reflecting the reduced allowance).
     * @param account The account whose tokens will be burnt.
     * @param value The amount that will be burnt.
     */
    function _burnFrom(address account, uint256 value) internal {
        _burn(account, value);
        _approve(account, msg.sender, _allowed[account][msg.sender].sub(value));
    }
}

// File: ERC20Burnable.sol



pragma solidity ^0.5.7;

import "./ERC20.sol";

/**
 * @title Burnable Token
 * @dev Token that can be irreversibly burned (destroyed).
 */
contract ERC20Burnable is ERC20 {
    /**
     * @dev Burns a specific amount of tokens.
     * @param value The amount of token to be burned.
     */
    function burn(uint256 value) public {
        _burn(msg.sender, value);
    }

    /**
     * @dev Burns a specific amount of tokens from the target address and decrements allowance.
     * @param from address The account whose tokens will be burned.
     * @param value uint256 The amount of token to be burned.
     */
    function burnFrom(address from, uint256 value) public {
        _burnFrom(from, value);
    }
}

// File: ERC20Detailed.sol



pragma solidity ^0.5.7;

import "./IERC20.sol";

/**
 * @title ERC20Detailed token
 * @dev The decimals are only for visualization purposes.
 * All the operations are done using the smallest and indivisible token unit,
 * just as on Ethereum all the operations are done in wei.
 */
contract ERC20Detailed is IERC20 {
    string private _name;
    string private _symbol;
    uint8 private _decimals;

    constructor (string memory name, string memory symbol, uint8 decimals) public {
        _name = name;
        _symbol = symbol;
        _decimals = decimals;
    }

    /**
     * @return the name of the token.
     */
    function name() public view returns (string memory) {
        return _name;
    }

    /**
     * @return the symbol of the token.
     */
    function symbol() public view returns (string memory) {
        return _symbol;
    }

    /**
     * @return the number of decimals of the token.
     */
    function decimals() public view returns (uint8) {
        return _decimals;
    }
}

// File: IERC20.sol



pragma solidity ^0.5.7;

/**
 * @title ERC20 interface
 * @dev see https://eips.ethereum.org/EIPS/eip-20
 */
interface IERC20 {
    function transfer(address to, uint256 value) external returns (bool);

    function approve(address spender, uint256 value) external returns (bool);

    function transferFrom(address from, address to, uint256 value) external returns (bool);

    function totalSupply() external view returns (uint256);

    function balanceOf(address who) external view returns (uint256);

    function allowance(address owner, address spender) external view returns (uint256);

    event Transfer(address indexed from, address indexed to, uint256 value);

    event Approval(address indexed owner, address indexed spender, uint256 value);
}

// File: SafeMath.sol



pragma solidity ^0.5.7;

/**
 * @title SafeMath
 * @dev Unsigned math operations with safety checks that revert on error.
 */
library SafeMath {
    /**
     * @dev Multiplies two unsigned integers, reverts on overflow.
     */
    function mul(uint256 a, uint256 b) internal pure returns (uint256) {
        // Gas optimization: this is cheaper than requiring 'a' not being zero, but the
        // benefit is lost if 'b' is also tested.
        // See: https://github.com/OpenZeppelin/openzeppelin-solidity/pull/522
        if (a == 0) {
            return 0;
        }

        uint256 c = a * b;
        require(c / a == b);

        return c;
    }

    /**
     * @dev Integer division of two unsigned integers truncating the quotient, reverts on division by zero.
     */
    function div(uint256 a, uint256 b) internal pure returns (uint256) {
        // Solidity only automatically asserts when dividing by 0
        require(b > 0);
        uint256 c = a / b;
        // assert(a == b * c + a % b); // There is no case in which this doesn't hold

        return c;
    }

    /**
     * @dev Subtracts two unsigned integers, reverts on overflow (i.e. if subtrahend is greater than minuend).
     */
    function sub(uint256 a, uint256 b) internal pure returns (uint256) {
        require(b <= a);
        uint256 c = a - b;

        return c;
    }

    /**
     * @dev Adds two unsigned integers, reverts on overflow.
     */
    function add(uint256 a, uint256 b) internal pure returns (uint256) {
        uint256 c = a + b;
        require(c >= a);

        return c;
    }

    /**
     * @dev Divides two unsigned integers and returns the remainder (unsigned integer modulo),
     * reverts when dividing by zero.
     */
    function mod(uint256 a, uint256 b) internal pure returns (uint256) {
        require(b != 0);
        return a % b;
    }
}

// File: Token.sol



pragma solidity ^0.5.7;

import "./ERC20.sol";
import "./ERC20Detailed.sol";
import "./ERC20Burnable.sol";

 
contract Token is ERC20, ERC20Detailed, ERC20Burnable {
    uint8 public constant DECIMALS = 18;
    // uint256 public constant INIT_SUPPLY = 10000000000 * (10 ** uint256(DECIMALS));
     
    // address private constant ADDR_TOKEN_SWAP = address(0x1234);
    // address private constant ADDR_OPERATION = address(0x1234);
    // address private constant ADDR_REWARD = address(0x1234);
    // address private constant ADDR_TEAM = address(0x1234);
    // address private constant ADDR_PARTNER = address(0x1234);
    
    // 2019.06.01 ~ 2021.12.01.
    uint[31] private TIME_TABLE = [ 1559347200,
        1561939200, 1564617600, 1567296000, 1569888000, 1572566400, 1575158400, 
        1577836800, 1580515200, 1583020800, 1585699200, 1588291200, 1590969600, 
        1593561600, 1596240000, 1598918400, 1601510400, 1604188800, 1606780800,
        1609459200, 1612137600, 1614556800, 1617235200, 1619827200, 1622505600, 
        1625097600, 1627776000, 1630454400, 1633046400, 1635724800, 1638316800];
        
    
    // event Mint(address indexed to, uint256 amount);
    // event FininshedMint();

    struct TimeMint {
        address beneficiary;
        uint256 releaseTime;
        uint256 value;
    }
    
    TimeMint[] timeMintList;

    address owner;
    bool mintingAvail = true;
    
    modifier onlyOwner() {
        require(msg.sender == owner);
        _;
    }
    
    modifier canMint() {
        require(mintingAvail);
        _;
    }

    /**
     * @dev Constructor that gives msg.sender all of existing tokens.
     */
    constructor () public ERC20Detailed("HAMA token", "HAMA", DECIMALS) {
        
        // All tokens will provide by insertTimeMintTimeTable
        // _mint(msg.sender, INITIAL_SUPPLY);
        owner = msg.sender;
        // _initMintScheuld();
    }
    
    /**
     * @dev Initialize supplying tokens. But This code occuers an error: out of gas
     */
    // function _initMintScheuld() internal{
        // insertTimeMint(ADDR_TOKEN_SWAP, TIME_TABLE[0], 1500000000);
        // insertTimeMint(ADDR_TOKEN_SWAP, TIME_TABLE[2], 200000000);
        // insertTimeMint(ADDR_TOKEN_SWAP, TIME_TABLE[3], 100000000);
        // insertTimeMint(ADDR_TOKEN_SWAP, TIME_TABLE[4], 300000000);
        // insertTimeMint(ADDR_TOKEN_SWAP, TIME_TABLE[5], 100000000);
        // insertTimeMint(ADDR_TOKEN_SWAP, TIME_TABLE[6], 300000000);
        // insertTimeMint(ADDR_TOKEN_SWAP, TIME_TABLE[7], 100000000);
        // insertTimeMint(ADDR_TOKEN_SWAP, TIME_TABLE[8], 300000000);
        // insertTimeMint(ADDR_TOKEN_SWAP, TIME_TABLE[9], 100000000);
        // insertTimeMint(ADDR_TOKEN_SWAP, TIME_TABLE[10], 300000000);
        // insertTimeMint(ADDR_TOKEN_SWAP, TIME_TABLE[11], 100000000);
        // insertTimeMint(ADDR_TOKEN_SWAP, TIME_TABLE[12], 100000000);
        
        // for (uint i=0; i<10; i+=3) {
        //     insertTimeMint(ADDR_OPERATION, TIME_TABLE[i], 375000000);
        // }
        
        // insertTimeMint(ADDR_REWARD, TIME_TABLE[0], 3000000000);
        
        // for (uint i=7; i<25; i++) {
        //     insertTimeMint(ADDR_TEAM, TIME_TABLE[i], 38920000);
        // }
        
        // for (uint i=0; i<25; i+=3) {
        //     insertTimeMint(ADDR_PARTNER, TIME_TABLE[i], 162500000);
        // }
    // }

    /**
     * @dev Getting the number of release tokens schedules.
     */
    function getTimeTimeLength() view public returns(uint256) {
        return timeMintList.length;
    }

    /**
     * @dev Getting the number of release tokens schedules.
     */
    function getTimeTimeMint(uint index) view public returns(address, uint256, uint256) {
        return (timeMintList[index].beneficiary, timeMintList[index].releaseTime, timeMintList[index].value);
    }

    /**
     * @dev Insert a minting tokens schedule. This method use index of TimeTable instead of timestamp.
     */
    function insertTimeMintTimeTable(address _beneficiary, uint256 index, uint256 _value) onlyOwner canMint public returns (bool){
        require(_beneficiary != address(0));
        require(index >= 0 && index < TIME_TABLE.length, "insertTimeMint: index out of range");
        return insertTimeMint(_beneficiary, TIME_TABLE[index], _value);
    }
    
    /**
     * @dev Insert a minting tokens schedule.
     */
    function insertTimeMint(address _beneficiary, uint256 _releaseTime, uint256 _value) onlyOwner canMint public returns (bool){
        require(_beneficiary != address(0));
        require(_releaseTime > block.timestamp, "TokenTimelock: release time is before current time");
        TimeMint memory item = TimeMint({
            beneficiary: _beneficiary,
            releaseTime: _releaseTime,
            value: _value
        });
        timeMintList.push(item);
        return true;
    }
    
    /**
     * @dev Remove a minting tokens schedule with index. The timeMintList length reduce one because "delete" is set value to empty.
     */
    function removeTimeMint(uint index) onlyOwner canMint public returns (bool){
        require(index >= 0 && index < timeMintList.length, "removeTimeMint: index out of range.");
        timeMintList[index] = timeMintList[timeMintList.length-1];
        delete timeMintList[timeMintList.length-1];
        timeMintList.length--;
        return true;
    }
    
 /**
     * @dev Releasing tokens any scheduled over release Time.
     */
    function releaseTimeMintToken() onlyOwner canMint public returns (bool) {
        require(timeMintList.length > 0);
        uint i = 0;
        while (i < timeMintList.length) {
            if (block.timestamp >= timeMintList[i].releaseTime) {           // 0. if True
                if (_mintForTime(timeMintList[i])) {                        // 1. Mining according to timeMintList[i].
                    timeMintList[i] = timeMintList[timeMintList.length-1];  // 2.  Note: timeMintList[timeMintList.length-1] is timeMintList[i] NOW !!
                    delete timeMintList[timeMintList.length-1];
                    timeMintList.length--;
                    continue;   // To solve the problem.
                }
            }
            i++;    // 3. BUT we skipped the check of timeMintList[i](i.e. timeMintList[timeMintList.length-1]) with `i++`, 
        }
    }
    
    function _mintForTime(TimeMint memory item) onlyOwner canMint private returns (bool) {
        require(block.timestamp >= item.releaseTime, "TokenTimelock: current time is before release time");
        _mint(item.beneficiary, item.value);
        // emit Mint(item.beneficiary, item.value);
        return true;
    }
    
    /**
     * @dev block mint function.
     */
    function finishMinting() onlyOwner canMint public returns (bool) {
        mintingAvail = false;
        // emit FininshedMint();
        return true;
    }

}